Permutations and Permutation Groups

```
> restart:with(GroupTheory):with(plots):
```

We create a permutation.

```
> a:=Perm([[1,2,15,6],[3,4,5]]);
```

$$a := (1, 2, 15, 6)(3, 4, 5)$$

We find the item the permutation maps 15 to.

```
> a[15];
```

$$6$$

The **degree** of the permutation is the largest number it changes.

```
> PermDegree(a);
```

$$15$$

We find the order of the permutation.

```
> PermOrder(a);
```

$$12$$

We use a loop structure to find the powers of the permutation *a*.

```
> for i from 1 to 12 do
  power[i]=PermPower(a,i)
  end do;
```

$$power_1 = (1, 2, 15, 6)(3, 4, 5)$$

$$power_2 = (1, 15)(2, 6)(3, 5, 4)$$

$$power_3 = (1, 6, 15, 2)$$

$$power_4 = (3, 4, 5)$$

$$power_5 = (1, 2, 15, 6)(3, 5, 4)$$

$$power_6 = (1, 15)(2, 6)$$

$$power_7 = (1, 6, 15, 2)(3, 4, 5)$$

$$power_8 = (3, 5, 4)$$

$$power_9 = (1, 2, 15, 6)$$

$$power_{10} = (1, 15)(2, 6)(3, 4, 5)$$

$$power_{11} = (1, 6, 15, 2)(3, 5, 4)$$

$$power_{12} = (\ )$$

A shortcut for finding a particular power.

```
> g:=a^7;
```

$$g := (1, 6, 15, 2)(3, 4, 5)$$

We see that 8 is fixed by the permutation.

```
> g[8];
```

$$8$$

An even permutation has a **parity** of 1, while an odd permutation has a parity of -1.

```
> PermParity(a);
```

$$-1$$

We define a second permutation.

```
> b:=Perm([[1,3],[2,6]]);
```

$$b := (1, 3)(2, 6)$$

We find its order.

```
> PermOrder(b);
```

$$2$$

We find its parity.

```
> PermParity(b);
```

$$1$$

We form the product of *a* and *b*.

```
> c:=a.b;
```

$$c := (1, 6, 3, 4, 5)(2, 15)$$

We find what *c* maps 3 to.

```
> c[3];
```

$$4$$

The order of *c*.

```
> PermOrder(c);
```

$$10$$

The parity of *c*.

```
> PermParity(c);
```

$$-1$$

We find the product of *b* and *a*.

```
> d:=b.a;
```

$$d := (1, 4, 5, 3, 2)(6, 15)$$

We find what *d* maps 3 to.

```
> d[3];
```

$$2$$

Thus this multiplication is not necessarily commutative. We find the order of *c*.

```
> PermOrder(c);
```

$$10$$

Next we enter the symmetric group on 4 symbols.

```
> G:=SymmetricGroup(4);
```

$$G := \mathbf{S}_4$$

We find its order.

```
> GroupOrder(G);
```

$$24$$

Is it Abelian?

```
> IsAbelian(G);
```

$$\mathit{false}$$

Is it cyclic?

```
> IsCyclic(G);
```

false

What are its 24 elements?

```
> E:=Elements(G);
```

$E := \{(\ ), (1,2), (1,3), (1,4), (2,3), (2,4), (3,4), (1,2,3), (1,2,4), (1,3,2), (1,3,4), (1,4,2), (1,4,3), (2,3,4), (2,4,3), (1,2,3,4), (1,2,4,3), (1,3,2,4), (1,3,4,2), (1,4,2,3), (1,4,3,2), (1,2)(3,4), (1,3)(2,4), (1,4)(2,3)\}$

We enter the Alternating group on 4 symbols.

```
> A:=AlternatingGroup(4);
```

$A := \mathbf{A}_4$

We find its order.

```
> GroupOrder(A);
```

12

What are its 12 elements?

```
> EE:=Elements(A);
```

$EE := \{(\ ), (1,2,3), (1,2,4), (1,3,2), (1,3,4), (1,4,2), (1,4,3), (2,3,4), (2,4,3), (1,2)(3,4), (1,3)(2,4), (1,4)(2,3)\}$

Is it Abelian?

```
> IsAbelian(A);
```

false

Is A a subgroup of G?

```
> IsSubgroup(A,G);
```

true

We define another permutation group by giving its generators.

```
> g1 := PermutationGroup({[[1, 2]], [[1, 2, 3], [4, 5]]});
```

$g1 := \langle (1,2), (1,2,3)(4,5) \rangle$

The order of the group.

```
> GroupOrder(g1);
```

12

Its elements.

```
> Elements(g1);
```

$\{(\ ), (1,2), (1,3), (2,3), (4,5), (1,2,3), (1,3,2), (1,2)(4,5), (1,3)(4,5), (2,3)(4,5), (1,2,3)(4,5), (1,3,2)(4,5)\}$

We draw a Cayley table for the group.

```
> DrawCayleyTable(g1);
```



	e	a	b	c	d	f	g	h	i	j	k	l
e	e	a	b	c	d	f	g	h	i	j	k	l
a	a	e	l	k	j	i	h	g	f	d	c	b
b	b	l	e	d	c	g	f	i	h	k	j	a
c	c	k	f	h	b	j	e	a	d	l	g	i
d	d	j	g	i	e	k	b	l	c	a	f	h
f	f	i	c	b	h	e	j	d	a	g	l	k
g	g	h	d	e	i	b	k	c	l	f	a	j
h	h	g	j	a	f	l	c	k	b	i	e	d
i	i	f	k	l	g	a	d	j	e	h	b	c
j	j	d	h	f	a	c	l	b	k	e	i	g
k	k	c	i	g	l	d	a	e	j	b	h	f
l	l	b	a	j	k	h	i	f	g	c	d	e

The center of the group is the upper left corner. Next we redraw the table with the literal permutations as labels.

```
> DrawCayleyTable(g1,'labels=literal',size=[1500,1500]);
```


( )
(4, 5)
(1, 2)
(1, 2, 3)(4, 5)

We create a group structure for the center of the group.

```
> g1c:=Center(g1);
```

$$g1c := Z(\langle (1, 2), (1, 2, 3)(4, 5) \rangle)$$

We list the elements of the center.

```
> Elements(g1c);
```

$$\{(\ ), (4, 5)\}$$

We give the generators for a subgroup of *g1*.

```
> g2 := Subgroup({[[1, 3, 2], [4, 5]]}, g1);
```

$$g2 := \langle (1, 3, 2)(4, 5) \rangle$$

We check that it is a subgroup.

```
> IsSubgroup(g2,g1);
```

true

We find the order of the subgroup.

```
> GroupOrder(g2);
```

6

We list its elements.

```
> Elements(g2);
```

$$\{(\ ), (4, 5), (1, 2, 3), (1, 3, 2), (1, 2, 3)(4, 5), (1, 3, 2)(4, 5)\}$$

We draw the Cayley table for the subgroup.

```
> DrawCayleyTable(g2,'labels=literal',size=[750,750]);
```

	( )	(1, 3, 2)(4, 5)	(1, 2, 3)	(4, 5)	(1, 3, 2)	(1, 2, 3)(4, 5)
( )	( )	(1, 3, 2)(4, 5)	(1, 2, 3)	(4, 5)	(1, 3, 2)	(1, 2, 3)(4, 5)
(1, 3, 2)(4, 5)	(1, 3, 2)(4, 5)	(1, 2, 3)	(4, 5)	(1, 3, 2)	(1, 2, 3)(4, 5)	( )
(1, 2, 3)	(1, 2, 3)	(4, 5)	(1, 3, 2)	(1, 2, 3)(4, 5)	( )	(1, 3, 2)(4, 5)
(4, 5)	(4, 5)	(1, 3, 2)	(1, 2, 3)(4, 5)	( )	(1, 3, 2)(4, 5)	(1, 2, 3)
(1, 3, 2)	(1, 3, 2)	(1, 2, 3)(4, 5)	( )	(1, 3, 2)(4, 5)	(1, 2, 3)	(4, 5)
(1, 2, 3)(4, 5)	(1, 2, 3)(4, 5)	( )	(1, 3, 2)(4, 5)	(1, 2, 3)	(4, 5)	(1, 3, 2)

```
> g2c:=Center(g2);
```

$$g2c := Z(\langle (1, 3, 2)(4, 5) \rangle)$$

We notice that the center is the entire subgroup, so the subgroup is Abelian.

```
> IsAbelian(g2);
```

true

```
> GG:=DihedralGroup(4,form = "permgroup");
```

$$GG := \mathrm{D}_4$$

```
> DrawCayleyTable(GG,'labels=literal',size=[800,800]);
```


( )
(1, 3)(2, 4)
(1, 3)
(1, 2, 3, 4)
(1, 4)(2, 3)
(1, 2)(3, 4)
(1, 4, 3, 2)
(2, 4)